a characterization of OMEGA

Johan G. F. Belinfante
2007 September 6

```
In[1]:= SetDirectory["l:"]; << goedel97.06a; << tools.m

        :Package Title: goedel97.06a           2007 September 6 at 3:20 p.m.

        It is now:  2007 Sep 6 at 21:14

        Loading Simplification Rules

        TOOLS.M                                     Revised 2007 June 25

        weightlimit = 40
```

summary

The class **OMEGA**: of all ordinal numbers is successor-invariant and closed under arbitrary unions.

```
In[2]:= Uclosure[OMEGA]

Out[2]= OMEGA

In[3]:= invariant[SUCC, OMEGA]

Out[3]= True
```

These two properties can be used to characterize the class of ordinals. It is shown in this notebook that the class **OMEGA** is the smallest class which is successor-invariant and closed under arbitrary unions. Comment: This result holds independently of the axiom of regularity.

derivation

Lemma. (For convenience, the temporary abbreviation **t** is used here for the least ordinal that does not belong to **x**. This temporary variable is eliminated from the final statement of the lemma.)

```
In[4]:= (Map[not, SubstTest[and, implies[and[p3, p4], p5],
          implies[p5, p7], implies[and[p4, p7], p8], implies[and[p5, p8], p9],
          not[implies[and[p3, p4], p9]], {p3 → not[subclass[OMEGA, x]],
           p4 → equal[t, A[dif[OMEGA, x]]], p5 → member[t, OMEGA],
           p6 → not[member[t, x]], p7 → subclass[t, OMEGA], p8 → subclass[t, x],
           p9 → member[U[t], Uclosure[x]]}]] // Reverse) /. t → A[dif[OMEGA, x]]

Out[4]= or[equal[core[x, U[A[intersection[OMEGA, complement[x]]]]],
          U[A[intersection[OMEGA, complement[x]]]]], subclass[OMEGA, x]] == True

In[5]:= (% /. x → x_) /. Equal → SetDelayed
```

Lemma.

```
In[6]:= (Map[not, SubstTest[and, implies[and[p3, p4], p9], implies[and[p1, p5, p9], q1],
          not[implies[and[p1, p3, p4], q1]], {p1 -> equal[x, Uclosure[x]],
           p3 → not[subclass[OMEGA, x]], p4 → equal[t, A[dif[OMEGA, x]]],
           p5 → member[t, OMEGA], p9 → equal[core[x, U[t]], U[t]],
           q1 → member[U[t], x]}]] // Reverse) /. t → A[dif[OMEGA, x]]

Out[6]= or[member[U[A[intersection[OMEGA, complement[x]]]], x],
         not[equal[x, Uclosure[x]]], subclass[OMEGA, x]] == True

In[7]:= (% /. x → x_) /. Equal → SetDelayed
```

Lemma.

```
In[8]:= (Map[not, SubstTest[and, implies[and[p3, p4], p5],
          implies[and[p3, p4], p6], implies[and[p1, p3, p4], q1], implies[and[p6, q1], q2],
          not[implies[and[p1, p3, p4], q3]], {p1 -> equal[x, Uclosure[x]], p3 →
            not[subclass[OMEGA, x]], p4 → equal[t, A[dif[OMEGA, x]]], p5 → member[t, OMEGA],
           p6 → not[member[t, x]], q1 → member[U[t], x], q2 → not[equal[U[t], t]],
           q3 → equal[t, succ[U[t]]]}]] /. t → A[dif[OMEGA, x]]) // Reverse

Out[8]= or[equal[A[intersection[OMEGA, complement[x]]],
          succ[U[A[intersection[OMEGA, complement[x]]]]]],
         not[equal[x, Uclosure[x]]], subclass[OMEGA, x]] == True

In[9]:= (% /. x → x_) /. Equal → SetDelayed
```

Main Theorem. The class of ordinals is the least class that is successor-invariant and closed under arbitrary unions.

```
In[10]:= Map[not, SubstTest[and, implies[and[p1, p3], q3], implies[and[p2, q1], q4],
          not[implies[and[p1, p2], not[p3]]], {p1 → equal[x, Uclosure[x]],
           p2 → subclass[image[SUCC, x], x], p3 → not[subclass[OMEGA, x]],
           q1 → member[U[A[intersection[OMEGA, complement[x]]]], x],
           q3 → equal[A[intersection[OMEGA, complement[x]]],
             succ[U[A[intersection[OMEGA, complement[x]]]]]],
           q4 → member[succ[U[A[intersection[OMEGA, complement[x]]]]], x]}]] // Reverse

Out[10]= or[not[equal[x, Uclosure[x]]],
          not[subclass[image[SUCC, x], x]], subclass[OMEGA, x]] == True
```

```
In[11]:= or[not[equal[x_, Uclosure[x_]]],
           not[subclass[image[SUCC, x_], x_]], subclass[OMEGA, x_]] := True
```

Corollary. There is no set that is successor-invariant and closed under arbitrary unions.

```
In[12]:= Map[equal[V, #] &, SubstTest[class, x, implies[member[x, u], subclass[v, x]],
           {u → intersection[invar[SUCC], fix[UCLOSURE]], v → OMEGA}]]

Out[12]= equal[0, intersection[fix[UCLOSURE], invar[SUCC]]] == True
```

```
In[13]:= intersection[fix[UCLOSURE], invar[SUCC]] := 0
```

comments and acknowledgments

The theorem in this notebook was proposed (on page 208) as a definition of the class of ordinal numbers by Jan Mycielski.

```
"Jan Mycielski, A System of Axioms of Set Theory for the Rationalists, Notices of the
  American Mathematical Society, Volume 53, Number 2, pages 206–213, February 2006."; "
```

A result somewhat similar to the final corollary in the preceding section was derived on 2005 January 15 in another notebook. Benjamin Lamothe, then a student at the Massachusetts Institute of Technology, and the present author used the **GOEDEL** program to derive what is commonly called Hilbert's Paradox (1905): there is no set that is closed under arbitrary unions and the power set operation:

```
In[14]:= disjoint[fix[UCLOSURE], invar[POWER]]

Out[14]= True
```